
Overview

The word “fractal” is often used in referring any object that is recursively constructed so that it appears
similar at all scales of magnification. There are many examples of complex real-life phenomena, such
as chaos, ferns, mountains, river networks, biological growth, that can be described and studied using
fractals. In this lab and project, we use the Maple to analyze and to generate a classic fractal, the
Koch snowflake, and its variations.

Koch Snowflake

1. Basic Construction: It starts with an equilateral triangle. A smaller equilateral triangle is
then added to each of the three sides. It is done in such a way that the base of each new triangle
is the middle one-third of each side of the original figure. This process is repeated again and
again in each successive iteration. Notice that we are getting more and more sides but the basic
construction is the same: each side is replaced by four sides of one-third of its length. The Koch
snowflake is the limit of this construction.

2. Working with the Maple:

(a) As always, we start a Maple session with restart and needed packages.
> restart;
> with(plots):

(b) Initial Data: Each side of the initial equilateral triangle is determined by two of its endpoints
(vertices) so we need a sequence of four points with the first and the last being the same
vertex.
> A[1]:=[0,0];
> A[2]:=[1/2,sqrt(3)/2];
> A[3]:=[1,0];
> A[4]:=A[1];

(c) Initial Results (level 1): The following should be obvious. Notice that plot automatically
connects a given sequence of points. Ask your TA if you want to know more about the
sprintf option in plot.
> LenSide[1] := 1;
> TotArea[1] := LenSide[1]*sqrt(3.)/2.*LenSide[1]/2.;
> NumSide[1] := 3;
> TotPerim[1] := NumSide[1]*LenSide[1];
> plot([A[1],A[2],A[3],A[4]],axes=NONE, scaling=constrained, title=sprintf
("Koch snowflake (level 1) \n area = % 8.5f, perimeter = % 8.5f, sides =% d",
TotArea[1], TotPerim[1], NumSide[1]));

(d) Results up to level k and limits: Please justify the formulas and limits in your project report.
> k:=10:
> for n from 1 to k-1 do
> level:=n+1;
> LenSide[n+1] := LenSide[n]/3.;
> TotArea[n+1] := TotArea[n]+NumSide[n]*TotArea[1]/9.^n;
> NumSide[n+1] := 4*NumSide[n];
> TotPerim[n+1] := NumSide[n+1]*LenSide[n+1];
> end do;
> TotAreaLimit:=sqrt(3.)*2/5;
> TotPerimLimit:=infinity;

(e) Construct sequences of snowflake vertices up to level k: Start from level 1, between each pair
of adjacent vertices A[m] and A[m + 1], we need to add three new vertices B[m], E(m), and
C[m]. We then have to re-index them to form a new sequence of vertices at the next level.
Pay close attention to your TA’s explanation about formulas that produce new vertices, as
you need to understand them to complete the project. Also, re-index the vertices explicitly
by hand for the first few levels to verify the general formulas. How many vertices are there
at level n?
> k:=6: # Computations for levels above 6 may require too much time and
memory for your computer to handle
> for n from 1 to k-1 do
> for m from 1 to 3*4^(n-1) do
> B[m]:=A[m]+(A[m+1]-A[m])/3;
># If let A[m+1]-A[m]=[x,y], then [-y,x]=[A[m][2]-A[m+1][2],A[m+1][1]-A[m][1]]
> E[m]:=A[m]+(A[m+1]-A[m])/2+sqrt(3)/6.*[A[m][2]-A[m+1][2],A[m+1][1]-A[m][1]];
> C[m]:=A[m]+2*(A[m+1]-A[m])/3;
> end do;
> for m from 1 to 3*4^(n-1) do
> Temp[4*(m-1)+1]:=simplify(A[m]);
> Temp[4*(m-1)+2]:=simplify(B[m]);
> Temp[4*(m-1)+3]:=simplify(E[m]);
> Temp[4*(m-1)+4]:=simplify(C[m]);
> end do;
> for m from 1 to 3*4^n do
> A[m]:=Temp[m];
> end do;
> A[3*4^n+1]:=A[1];
># Construct sequence of snowflake plots
> SFplots[n+1] := seq(plot([A[m],A[m+1]]),m=1..3*4^n):
> end do:

(f) Display the fractal growth
> for n from 2 to k do
> display([SFplots[n]],axes=NONE, scaling=constrained, title=sprintf
("Koch snowflake (level %a) \n area = %8.5f, perimeter = %8.5f, sides =%d",
n,TotArea[n], TotPerim[n], NumSide[n]));
> end do;

Remark: While the for-loop is easy to follow, it is not very efficient in computing. For those who are
interested, a link to a faster Maple program for the same construction is included on the lab website.

Instructions: Your report should follow the guidelines set forth in the Project Report Guidelines on
the lab website and is due by the date specified by your TA. It should cover the following:

1. Study the given Koch snowflake construction and answer all questions asked. In particular, verify
the formulas and limits presented there using your knowledge in sequences and series. Have a
discussion on what you have learned, such as some interesting properties of the Koch snowflake.
For example, as we have computed, the Koch snowflake has a finite area but infinite perimeter.
Now, imaging that you have a container with the Koch snowflake as its base and fill it up with
some paint. That means one could paint an infinite area (the interior surface of the container)
with a finite amount of paint! The Koch snowflake is also an example of curves that is everywhere
continuous but nowhere differentiable.

2. Repeat the same discussion and analysis with your own constructions in Maple of at least two
different variations of the Koch snowflake. Use your imagination but one of them must be con-
structed as follows: Instead of using triangles, start with a square and recursively add smaller
squares. Notice that, once you understand the given construction of the Koch snowflake, it should
be easily modified for different variations.

